
    Prefácio

    
        
        
            Poucos anos atrás, eu me vi entrando no mundo do Clojure para resolver o problema de processar muitos dados diferentes ao mesmo tempo e fazer o merge deles em uma estrutura de dados só. Naquela época, era difícil encontrar coisas de Clojure em português e/ou com tanta qualidade quanto o livro do Gregório. Minha trajetória foi divertida, mas a partir de hoje vou certamente passar a indicar este livro como o primeiro contato com Clojure.

Quando o Gregório pediu para eu ler o livro e dar um feedback, tive de apagar minhas opiniões centenas de vezes, pois parecia que ele já tinha previsto o que eu ia falar. Foi uma dinâmica incrível ler pensando que seria legal adicionar algo, e encontrá-lo lá, no próximo capítulo. Creio que o livro tem todo o conteúdo que eu gostaria de ver em um livro de Clojure.

Agora, se você já sabe um pouco de Clojure, vá para a segunda e terceira partes e tente aplicar as ideias do livro em seu projeto ou dia a dia. Espero que você goste deste livro tanto quanto eu!

Julia Naomi Boeira

Este livro é para você

O intuito deste livro é ser um material introdutório à Programação Funcional. Mesmo sendo introdutório, ele não tem a intenção de ser o seu primeiro livro de programação. Portanto, saber programar um pouquinho será muito útil. Se você já escreveu algumas linhas de código na sua vida, este livro pode ser para você.

Aqui você aprenderá novos conceitos que vão ajudá-lo a escrever melhores programas de computador. Estes conceitos serão aprendidos ao longo do livro e, assim que apresentados, já podem ser aplicados no seu trabalho ou estudo. Tenho a expectativa de que, enquanto lê este livro, você procure aplicar tudo o que aprender, e demonstre para sua equipe as vantagens do que viu aqui. E, ainda, quando concluir a leitura, que você veja o tanto que terá se desenvolvido na escrita de software.

Veremos um bocado de código em Clojure. Se você tem a mente aberta para uma nova linguagem de programação e acredita que uma sintaxe e paradigma novos o levarão a novos horizontes, este livro é para você.

Por ser introdutório, não abordaremos conceitos como monads e applicatives, que têm assustado muita gente ao entrar neste mundo (inclusive eu mesmo), nem equações matemáticas.

Como este livro está organizado

Nossa jornada será muito interessante. Introduziremos princípios da Programação Funcional, e aos poucos veremos algumas comparações com linguagens historicamente consideradas orientadas a objeto (como Java e Ruby).

Na primeira parte do livro, veremos um pouco de Clojure, o suficiente para que nos familiarizemos com a linguagem. Isto deve facilitar a compreensão dos exemplos e conceitos de Programação Funcional, apresentados na parte seguinte, que é onde vamos conhecer e nos aprofundar em seus princípios. Algumas ideias muito bacanas aparecerão, trazendo um vocabulário novo para nós, como preguiça.

Na parte 3, veremos como construir aplicações em Clojure. Utilizando o domínio e as funções da parte 2, produziremos um serviço para processar requisições HTTP, fazendo uso de uma abordagem de desenvolvimento guiada por testes. E tudo isso acompanhado de muitos exemplos de código.

Participe da discussão

Os códigos de exemplo utilizados neste livro podem ser encontrados em https://gitlab.com/programacaofuncional/.

Sobre o autor

Natural de João Pessoa, Paraíba, Gregório Melo é desenvolvedor de software na ThoughtWorks Brasil. Estudou Desenvolvimento de Sistemas para Internet no IFPB e iniciou a carreira na área de desenvolvimento de software estagiando em uma grande empresa nacional de varejo, gerando Strings imensas a partir de um formulário gigante para criação de cartões de crédito.

Em 2011, mudou-se para Porto Alegre, onde teve seus primeiros contatos com Programação Funcional. Após um período de contatos esporádicos com linguagens funcionais, foi durante um engajamento profissional em Xi'an, na China, que Programação Funcional virou presença constante no seu dia a dia. O foco, na época, era Scala. Hoje, dedica-se a Clojure, sem querer discutir qual das duas é melhor. Foi gerente do escritório da ThoughtWorks em São Paulo entre junho de 2016 e dezembro de 2017, e hoje é, de novo, consultor de desenvolvimento na mesma.

Adora jogar futebol, mesmo com um talento um tanto quanto questionável, e talvez passe mais tempo do que deveria jogando videogames.

Agradecimentos

Este livro é um dos frutos de todo o carinho e amor que Diana e eu nutrimos entre nós. Obrigado por todo o incentivo e apoio! Não existiria livro algum sem você por perto.

Dentre os diversos privilégios que tenho, considero o mais significante ter convivido num lar onde a educação sempre foi muito valorizada. Agradeço a minha mãe e a meu pai por me proporcionar uma educação diferenciada, e a minha irmã e irmão pelo respeito e compaixão que temos entre nós.

Agradeço à Julia Naomi pela revisão técnica e pelo prefácio que me deixou emocionado. Quanto conhecimento! Também ao Paulo Caroli pela mentoria e todo o incentivo com o livro e tudo mais!

À Vivian, minha editora, que tem uma paciência exemplar! Vivian, obrigado por me acompanhar nessa jornada.

E gostaria de agradecer a todos os computadores que já rodaram os programas que eu criei. E também àqueles que pelo menos tentaram.


        
        

    



        
        
    
    


        
        

    Por que ler este livro?

    
        
        
            Se você se interessa em aprender a escrever programas melhores, Programação Funcional tem muito a lhe ajudar. Com ela, você pode produzir códigos mais robustos, menos suscetíveis a erros e expandir sua forma de pensar. Por exemplo, a imutabilidade, um conceito que veremos ao longo do livro, minimiza a possibilidade de encontrarmos defeitos oriundos da manipulação de estado em lugares desconhecidos. Você verá os principais conceitos que trazem à tona as vantagens da programação funcional.

Uma grande razão para aprender Programação Funcional é fazer uso dos benefícios do paralelismo, e veremos como fazer isso mais à frente. Escrever código paralelizável em uma linguagem de paradigma funcional é muito fácil. A ausência de efeitos colaterais (ou o estímulo para que eles sejam contidos e limitados) nas funções de um programa permite que estas funções sejam executadas sem uma ordem definida e em paralelo. Isso nos faz tirar proveito da nova aplicação da Lei de Moore, que agora não mais se refere à quantidade de transistores, mas sim de núcleos na máquina (http://www.gotw.ca/publications/concurrency-ddj.htm). Assim, nossa vida fica mais simples com a linguagem de programação nos ajudando. Você vai perceber como é fácil escrever código para rodar em múltiplas threads!

Ainda, para algumas pessoas, as origens na matemática fazem da Programação Funcional um paradigma mais simples de ser compreendido do que a Orientação a Objetos. Para estas pessoas, será muito natural a ideia de que, sempre que você chamar uma função com um mesmo parâmetro, o resultado será o mesmo, sem que nada mais no universo do programa seja alterado. Claro que isso não é exclusivo da Programação Funcional, já que podemos alcançar o mesmo resultado em outros paradigmas, mas a Programação Funcional vai ajudá-lo a pensar sempre em construir funções sem efeitos colaterais.

Por fim, talvez você queira entender um pouco mais sobre como escrever melhor código na plataforma que você utiliza atualmente. As ideias que foram citadas nos parágrafos anteriores podem ajudar seus programas a serem mais robustos. Você pode fazer com que seu código em JavaScript fique mais idiomático, e, de brinde, você vai entender melhor o processo de escrita de código para frameworks JavaScript como o D3, Ramda, Lodash, ou tantos outros. Seu código em Java 8, por exemplo, pode ficar muito mais sucinto e com melhor desempenho, tendo mais facilidade no uso do paralelismo provido pela máquina virtual.

Por que Clojure?

Existem muitas linguagens de programação que trazem consigo os princípios de Programação Funcional. Muitas delas têm conseguido uma boa adoção no mercado, como Scala, Clojure e Elixir. E linguagens cujas origens estão atreladas à Orientação a Objetos, como Java, C#, Ruby e Python, já há algum tempo possuem recursos encontrados em Linguagens Funcionais.

Então, por que Clojure? Além de ser a minha linguagem favorita do momento, ela possui recursos interessantes para nos ajudar a manter o foco nos aspectos inerentes à linguagem e à Programação Funcional: a sintaxe, que é simples e muito diferente das linguagens mais populares, e um sistema dinâmico de tipos, o que vai nos permitir evitar pensar em Orientação a Objetos por um tempo.

Clojure é um dialeto de Lisp (https://pt.wikipedia.org/wiki/Lisp). Lisp é uma linguagem que trouxe a ideia de que é possível utilizar apenas funções matemáticas para representar estruturas de dados básicas, aliado ao conceito de código como dado (do inglês Homoiconicity). Lisp tem muita história na computação, principalmente no mundo das linguagens funcionais e no campo da inteligência artificial. Sendo Clojure um dialeto de Lisp, ela traz uma sintaxe bem diferente das linguagens derivadas de Algol (como C e afins). Você vai perceber que, além de diferente do tradicional, a sintaxe é extremamente simples, com poucos símbolos ou palavras-chaves reservadas.

Clojure, principalmente, roda na máquina virtual Java, o que permite que programas escritos nesta linguagem usem bibliotecas escritas em Java. A ferramenta de automação de tarefas e gerenciamento de dependências, Leiningen, é bastante flexível e completa, e provê suporte ao repositório de bibliotecas do Maven. Ela também roda em outras plataformas, sendo JavaScript a mais notória (através do ClojureScript, que compila seu código Clojure para JavaScript), e no motor do Unity (para desenvolvimento de jogos) com o Arcadia (http://arcadia-unity.github.io/).

Existe uma comunidade muito legal por trás da linguagem, além de casos de sucesso de adoção de Clojure, que lhe dão um respaldo muito forte.

Fui convincente o bastante? Que tal começar a pôr a mão na massa e instalar as ferramentas necessárias?

Instalando Clojure

Você precisará instalar o kit de desenvolvimento do Java, JDK, na versão 8 ou 11. O OpenJDK deve bastar (http://openjdk.java.net). Se você já tem a JDK da Oracle instalada (versão 8 ou 11), não precisa se preocupar em instalar o OpenJDK.

Com o JDK instalado, basta instalar o Leiningen conforme instruções no site: http://leiningen.org/.

Para verificar se a instalação foi feita com sucesso, abra um novo terminal (ou prompt de comando), e digite lein -v. Deve aparecer uma mensagem dizendo qual é a versão instalada do Leiningen, o que significa que a instalação foi um sucesso. Talvez demore um pouco para o resultado deste comando aparecer, já que é a primeira execução do Leiningen e ele deve buscar na grande rede algumas bibliotecas essenciais para que nossos códigos em Clojure rodem com sucesso.


Neste link você vai encontrar um guia atualizado de como instalar Clojure no Windows, no Mac e no Linux: http://bit.ly/instalando-clojure.



Nosso primeiro código em Clojure (e o REPL)

Esta combinação de letras passará a ser comum no seu vocabulário, se já não for. REPL é um ambiente interativo onde escrevemos códigos e eles são interpretados de imediato, gerando resultados muito rápidos. É ideal para trechos pequenos e validação de ideias.

REPL é um acrônimo para read, evaluate, print e loop. Isto quer dizer que ele vai ler nossos comandos, interpretá-los, exibir na tela o resultado e repetir o processo. Se você programa em Ruby, talvez já tenha se deparado com irb ou pry. Python também oferece uma ferramenta para o mesmo propósito: é só digitar python no terminal e você terá um interpretador. Scala e Erlang são exemplos de linguagens cujas plataformas também oferecem tais ferramentas. Dica: o REPL do Clojure é o mais bacana.

Programar no REPL é algo bem comum entre quem tem familiaridade com Clojure. Claro, em algum momento nossos códigos vão parar em arquivos e serão empacotados. Mas o REPL fornece um ambiente muito prático para experimentação e testes. Algumas pessoas substituem desenvolvimento guiado por testes pelo constante uso do REPL. Eu prefiro continuar a criar casos de testes, e mais para a frente nós veremos como aplicar desenvolvimento guiado por testes aos nossos programas em Clojure. Agora, vejamos este tal de REPL na prática.

Em uma janela no terminal, digite:

lein repl

Um terminal interativo deve aparecer para você, com uma mensagem mais ou menos assim:

nREPL server started on port 59100 on host 127.0.0.1 - nrepl://127.0.0.1:59100

E, esperando você fornecer alguma informação, o terminal diz:

user=>

Isso significa que já podemos escrever código! Digite o seguinte:

"Valeu, Clojure!"


Aperte Enter e veja o texto Valeu, Clojure! aparecer na tela mais uma vez. Não sei você, mas eu acho que os passos para um Hello, world! como este foram práticos e rápidos (bem, talvez não tão rápidos por causa da primeira execução do Leiningen, quando ele baixa as dependências mínimas para começarmos a utilizar Clojure).

Digite exit para sair do REPL. Ou quit, ou aperte control + d. Já já voltaremos para ele, já que lá praticaremos as partes 1 e 2 do livro.

Ferramentas alternativas

O OpenJDK tem crescido bastante ao longo dos anos, sendo o suficiente para o desenvolvimento de aplicações em Clojure. Mas fique à vontade para utilizar a JDK da Oracle, se preferir.

O Leiningen é a ferramenta mais comum no meio Clojure. Mas existe uma alternativa que tem ganhado muita tração e interesse da comunidade, o Boot (http://boot-clj.com). Após acostumar-se com o Leiningen, recomendo que você pesquisa um pouco mais o Boot, que pode ser mais apropriado para você e/ou seu time.

Conclusão

Agora temos tudo que precisamos para metermos a mão na massa e começarmos a praticar Programação Funcional com Clojure! Estou aqui imaginando sua empolgação para correr para as próximas páginas. Nos vemos lá!


        
        

    



        
        
    
    


        
        


O suficiente de Clojure para começarmos a brincadeira




    Capítulo 1

Primeiros contatos com Clojure

    
        
        
            1.1 As primeiras linhas de código

Tudo preparado para escrevermos código loucamente? Agora que você já instalou as ferramentas necessárias, vamos praticar um pouco. No terminal, volte a executar o REPL:

lein repl

A ideia é vermos um pouco da sintaxe do Clojure para pegarmos um pouco de familiaridade, para, então, partirmos para coisas bem legais. Vamos ver algumas operações aritméticas primeiro. No REPL, digite a seguinte operação:

(+ 1 2)


O resultado, como você já deve estar esperando, é 3. Ufa, a máquina virtual ainda sabe fazer contas matemáticas. Mas a sintaxe parece bem diferente das linguagens com as quais estamos acostumados, não é? Vale vermos um pouco de explicação do que acontece na linha que acabamos de digitar.

Esta linha parece uma lista, composta de +, 1 e 2. E é isso mesmo. Acontece que o primeiro elemento desta lista é especial. Este elemento é uma função que é executada, e os demais elementos desta lista são argumentos para esta função. O mesmo se aplica a outras operações aritméticas:

(* 2 3)
(/ 2 2)
(- 0 2)


E se for uma operação um pouco mais complexa? Olhe só:

(* 2 (+ 3 3))
;; isto é um comentário em Clojure e vamos
;; utilizá-lo para demonstrar o retorno das funções
;; 12


Este exemplo é onde Clojure mostra mais um diferencial no sentido de sintaxe, fruto de sua herança de Lisp. Do ponto de vista de precedência, é claro que a soma será executada primeiro. O diferencial, porém, é que a linguagem exige os parênteses, o que não deixa margem alguma para dúvidas do que precede o quê. Portanto, a ordem de execução de código é sempre de dentro para fora. Então, primeiro veremos o código (+ 3 3) ser executado, e o resultado, 6, é argumento para o seguinte, (* 2 6).

Vamos ver como podemos concatenar Strings? Assim:

(str "Oi, " "mundo!")


Como podemos verificar se duas Strings são iguais? Assim:

(= "Oi" "Olá")
;; false

(= "Oi" "Oi")
;; true


É importante notar que, diferente de muitas linguagens, = é uma função em Clojure que verifica se duas coisas são iguais. O = não é um operador de associação, normalmente utilizado na construção de variáveis.

Vamos explorar mais. Como verificamos se um número é par?

(even? 2)
;; true


Se um número é múltiplo de 3?

(= 0 (mod 9 3))
;; true


Neste último exemplo, o que acontece é que vamos verificar o módulo da divisão entre 9 e 3. O resultado será utilizado como o segundo argumento na função que verifica igualdade, =.

1.2 Nossas próprias funções

Agora que já sabemos como chamar funções, que tal criarmos as nossas? Vamos começar do básico. Que tal uma função que receba um nome e dê um "oi"? Um exemplo de execução seria algo assim:

(oi "galera")


E o resultado esperado seria Oi, galera!.

Sabemos que precisamos criar uma função que se chame oi e que receba um só argumento, uma String. E que o coloque entre as Strings Oi, e !.

Para criar esta função, fazemos o seguinte:

(defn oi [nome]
  (str "Oi, " nome "!"))



Namespaces

Ao criar uma função no REPL, você verá que a saída é a seguinte:

#'user/oi

Isto quer dizer que alguma coisa com o nome oi acabou de ser criada, e encontra-se no namespace padrão, user. Namespaces em Clojure representam a mesma ideia que em outras linguagens, como pacotes em Java, sendo uma forma de agrupar funções. A combinação do namespace e do nome da função forma o identificador de tal função. 

A função +, por exemplo, é encontrada no namespace clojure.core, sendo seu identificador clojure.core/+. Como o namespace clojure.core é disponibilizado por padrão, a função + está sempre disponível. Funções em outros namespaces precisam ser incluídas no nosso código antes de serem utilizadas. Veremos como fazer isto em um outro momento.



Com a função criada, vamos aplicá-la ao nome desejado:

(oi "galera")
;; "Oi, galera!"

(oi "Margaret Hamilton")
;; "Oi, Margaret Hamilton!"


O defn nos indica que vamos criar uma função. Depois, damos um nome a ela; neste caso, oi. Logo a seguir, vem a lista de argumentos, cercada por [ e ]. Neste caso, temos apenas um só argumento, então fica [nome].

Em seguida vem o que realmente é executado: a concatenação de Strings. Note que, assim como em algumas linguagens, não precisamos definir o que será retornado. A última instrução é o que será retornado; neste caso, o resultado da concatenação. Note que o parêntese que foi aberto em defn só é fechado no final. Isto nos ajuda muito a entender até onde vai o escopo do código que criamos.

O trecho que concatena Strings tem uma particularidade: é aplicado em 3 argumentos. Esta é uma de várias funções que são aplicáveis em uma quantidade indeterminada de argumentos.

E como seria uma função que verifica se um número é múltiplo de 3? Precisamos verificar se, na divisão deste número por 3, o resto é 0.

(defn multiplo-de-3? [dividendo]
  (= 0 (mod dividendo 3)))


Então podemos chamar a função assim:

(multiplo-de-3? 9)
;; true

(multiplo-de-3? 20)
;; false


1.3 O que é a verdade? Sobre os condicionais

Há pouco aprendemos a criar nossas próprias funções e a colocar um pouco de lógica nelas. Mas e se quisermos dar um rumo diferente a esta lógica, dependendo de alguma condição? Clojure também tem condicionais (novidade!). Que tal incrementarmos nossa caixa de ferramentas e colocarmos uns condicionais à nossa disposição?

De novo, vamos aplicar algo simples na jogada. Que tal exibirmos a palavra sim se um número for par, e, caso contrário, a palavra não? Pegando carona com a função even?, nosso código fica assim:

(defn par? [numero]
  (if (even? numero)
    "sim"
    "não"))


Vamos ver como nos saímos?

(par? 5)
;; não

(par? 4)
;; sim


if tem muito a cara de uma função também. Até podemos enxergá-la como uma, mas ela é na verdade o que é chamado de forma especial: um recurso base do Clojure. Veremos outras (que são poucas, na verdade) ao longo do livro.

O if recebe 3 argumentos, começando com uma verificação que retorna verdadeiro ou falso (true ou false). Os demais argumentos representam algo a ser executado de acordo com o resultado da verificação. Em caso de verdadeiro, o argumento 2 é avaliado e retornado. Caso contrário, o argumento 3 é que é avaliado e retornado.

É importante salientar que apenas nil e false são considerados realmente falsos para a verificação de condições. Outros, como 0 e String vazia, que são comuns em outras linguagens, serão avaliados como verdadeiro!

Vale repetir: apenas nil e false são considerados realmente falsos!


Consultando a documentação

Sempre que você quiser saber mais sobre algum recurso do Clojure, você pode experimentar a documentação dentro do REPL! Para isso, use a função doc:

(doc if)



E se houver outra opção na nossa lógica, além de apenas duas? Por exemplo, dizer quando um saldo é positivo, negativo ou zerado. Nestes casos, a macro cond pode nos ajudar (veremos mais sobre macros no capítulo 5). Ela recebe pares de condicionais e expressões. Segue um exemplo:

(defn saldo [valor]
  (cond (< valor 0) "negativo"
        (> valor 0) "positivo"
        :else "zero"))


Vamos testar?

(saldo 900)
;; "positivo"

(saldo -900)
;; "negativo"

(saldo 0)
;; "zero"


Neste exemplo, quando o valor é menor que zero, o primeiro teste, logo de cara, retorna verdadeiro, e a expressão que o segue, "negativo", é avaliada (neste caso, não faz nada, apenas retorna a String "negativo"). Quando o valor é maior que zero, o primeiro teste ((< valor 0)) falha, retornando falso, mas o teste seguinte retorna verdadeiro, e a expressão correspondente é avaliada. Agora, quando o valor é exatamente 0, as duas primeiras verificações retornam falso, e a última verificação (:else) é validada e "zero" é retornado.

O que significa este :else? Nada demais, na real. Como qualquer coisa diferente de nil e false é considerada verdadeira, qualquer coisa que colocássemos ali no lugar do :else funcionaria como uma forma de fazer com que "zero" fosse retornado por padrão. Se quiser, teste lá com valores como 1 ou "milhão". Bem, não é realmente qualquer coisa: números (mesmo que 0), Strings (mesmo que vazias), caracteres, coleções... O :else é apenas uma convenção adotada por algumas pessoas as quais eu andei copiando. Veremos um pouco mais sobre essa "coisa" que começa com : no capítulo 4.

1.4 Conclusão

Ufa! Acho que agora temos a capacidade de aumentar um pouco mais a complexidade dos problemas que podemos tratar. Aqui aprendemos como interagir com o REPL do Clojure, chamar funções, criar nossas próprias funções e trabalhar com condicionais. No próximo capítulo, resolveremos um problema bastante comum no mundo da programação: o FizzBuzz. Você verá como a solução para este problema fica bem sucinta em Clojure.


        
        

    



        
        
    
    


        
        

    Capítulo 2

Resolvendo o FizzBuzz

    
        
        
            Com tudo o que aprendemos até aqui, podemos resolver um problema simples e bem comum em eventos de programação: o FizzBuzz (https://en.wikipedia.org/wiki/Fizz_buzz).

Explicando brevemente, devemos criar uma função que nos retorne:


	
fizz, se o número for divisível por 3;

	
buzz, se divisível por 5;

	
fizzbuzz, se divisível por 3 e por 5; ou

	o próprio número, caso contrário.



Quer tentar primeiro, antes que eu mostre uma possível solução?

Se não, vamos lá. Podemos começar pensando em como seria o nome ideal para uma função que fizesse o cálculo, e que recebesse um argumento, que é o número que buscamos. Que tal uma função que se chame fizzbuzz? Seria algo assim:

(defn fizzbuzz [numero]
  numero)


Desse jeito, chamaríamos a função assim: (fizzbuzz 8). Porém, claro, ainda não há lógica suficiente ali; o próprio número seria sempre retornado. Vamos primeiro criar uma função que verifica se um número é divisível por outro:

(defn divisivel-por? [dividendo divisor]
  (= 0 (mod dividendo divisor)))


Podemos dar uma melhorada de leve, substituindo o = 0 pela função nativa que verifica se um número é exatamente 0: zero?. Então a função de verificar divisibilidade fica assim:

(defn divisivel-por? [dividendo divisor]
  (zero? (mod dividendo divisor)))


Agora podemos chamar esta função dentro dos condicionais que checam a lógica do FizzBuzz, começando com a primeira verificação, a divisão por 3:

(defn fizzbuzz [numero]
  (cond (divisivel-por? numero 3)  "fizz"
        :else numero))


O primeiro passo foi dado. Agora podemos ver o resultado para os múltiplos de 3

(fizzbuzz 3)
;; "fizz"

(fizzbuzz 6)
;; "fizz"

(fizzbuzz 20)
;; 20

(fizzbuzz 8)
;; 8


Agora precisamos ver o caso dos múltiplos de 5:

(defn fizzbuzz [numero]
  (cond (divisivel-por? numero 3)  "fizz"
        (divisivel-por? numero 5)  "buzz"
        :else numero))


(fizzbuzz 3)
;; "fizz"

(fizzbuzz 6)
;; "fizz"

(fizzbuzz 20)
;; "buzz"

(fizzbuzz 8)
;; 8


E agora, e qual seria o resultado para o 15?

(fizzbuzz 15)
;; "fizz"


O resultado foi "fizz", o que é compreensível. O que acontece é que a primeira condição é válida, uma vez que 15 é divisível por 3. A saída é verificarmos, antes de tudo, se o número é divisível por 3 e por 5, como diz o enunciado do problema:

(defn fizzbuzz [numero]
  (cond (and (divisivel-por? numero 3)
             (divisivel-por? numero 5)) "fizzbuzz"
        (divisivel-por? numero 3)  "fizz"
        (divisivel-por? numero 5)  "buzz"
        :else numero))


Vamos testar?

(fizzbuzz 15)
;; "fizzbuzz"


Agora sim! A função and verifica a veracidade de todos os seus argumentos. Se todos eles forem true, and retorna true. Se algum deles for false, and retorna false. No caso do número 15, and recebe true e true, portanto, retorna true e "fizzbuzz" é o resultado.

Dado que já aprendemos sobre a função and, que tal praticar no REPL uma outra função interessante, a função or?

2.1 Conclusão

Esta é uma das várias possíveis soluções para este problema. Aqui vimos um pouco mais a aplicabilidade de condicionais e uma nova função, and. Logo mais daremos nosso primeiro passo no mundo funcional através da função map.


        
        

    



        
        
    
    


        
        

    Capítulo 3

Estruturas de dados em Clojure

    
        
        
            Agora que temos o FizzBuzz resolvido, que tal aplicar esta lógica a uma lista de números? Até agora vimos um pouco sobre estruturas de dados em Clojure e chegou a hora de vermos listas e mapas! Vejamos primeiro as listas.

3.1 Listas

;; é assim que criamos uma lista em Clojure
(list 1 2 3 4 5)
;; (1 2 3 4 5)

;; ou assim
'(1 2 3 4 5)
;; (1 2 3 4 5)



Por que o apóstrofo na construção da lista?

Bem observado! Lembre-se de que a primeira coisa que vem depois de um parêntese ( é normalmente uma função. E esta é uma característica de linguagens baseadas em Lisp, onde tudo é uma lista, inclusive quando aplicamos uma função. Sem o apóstrofo, a interpretação seria de que 1 seria compreendido como uma função, e seria aplicada aos demais argumentos. Com o apóstrofo, instruímos que a lista seja interpretada como uma simples lista, sem execução de função alguma.



Podemos até dar um nome a esta lista:

(def um-ate-5 '(1 2 3 4 5))
;; #'user/um-ate-5

(count um-ate-5)
;; 5


O def nos permite dar um nome a alguma coisa e esta coisa pode, então, ser referenciada mais adiante. É como a declaração de variáveis em outras linguagens, como o uso de const em JavaScript. E aí podemos passar o valor referenciado com def

